III olimpiada

I etapas

16. Kuriai metų savaitei priklauso diena. Šį uždavinį galima spręsti įvairiais būdais. Kadangi jis nesunkus, duomenų nėra daug, tuomet reikia pasirinkti aiškesnį ir trumpesnį algoritmą. 

Svarbiausia, ką reikia rasti šiam uždaviniui: 1) sužinoti, kelinta metų diena yra duotoji diena; 2) nustatyti pirmosios duotųjų metų savaitės trukmę. Po to apskaičiuoti savaitės numerį nesunku: tereikės pasinaudoti formule

(d_sk - trukmė - 1) div 7 + 2;

čia d_sk reiškia dienų skaičių nuo duotos datos metų pradžios iki pateiktos dienos, kitaip sakant, kelinta metų diena yra duotoji diena, o trukmė – pirmosios metų savaitės ilgis. Šį skirtumą sumažiname vienetu dėl to, kad septintoji savaitės diena (sekmadienis) priklauso tai pačiai savaitei (pagal lietuvišką kalendorių). Pavyzdžiui, jei sausio 7 d. yra sekmadienis, o ieškome kuriai savaitei priklauso sausio 14 diena, tai

(14 - 7 - 1) div 7 + 2 = 2

ir gauname, kad ieškoma diena yra antroje metų savaitėje. 

Dvejeto prisumavimas reikalingas tam, kad papildytume dviem savaitėmis: pirmąja metų savaite ir paskutine savaite, kurioje yra ieškoma data. Mat operacijos div rezultatas yra pilnų savaičių skaičius iki savaitės, kurioje yra duota diena (t .y. paskutinė savaitė neįtraukiama).

Nustatysime duotosios dienos numerį nuo metų pradžios. Tai įprastas datų skaičiavimo algoritmas. Laikome, kad data aprašyta įrašu. Nors neegzistuoja nulinė diena ar mėnuo, tačiau nulis įtraukiamas į atkarpos tipą, aprašantį mėnesį ir dieną. Mat atliekant skaičiavimus su datomis, kartais prireikia ir ribinės reikšmės – nulio.

type metai = 1583..5000;  { Grigaliaus kalendorius }
     data = record
              mt: metai;

              mėn: 0..12;

              diena: 0..31

            end;

function d_sk (dt: data): integer;

{ randa duotos dienos numerį nuo metų pradžios }
  var dienos: integer;

      t: data;

  function d_mėn (dt: data): integer;

  { kiek dienų turi duotų metų duotas mėnuo }
  begin
    case dt.mėn of
      4, 6, 9, 11: d_mėn := 30;

      2: if kel (dt.mt) { jei metai keliamieji }
            then d_mėn := 29

            else d_mėn := 28;

      1, 3, 5, 7, 

      8, 10, 12: d_mėn := 31

    end
  end;

begin
  dienos := 0;

  t := dt;

  for t.mėn := 1 to dt.mėn - 1 do
      dienos := dienos + d_mėn(t);

  d_sk := dienos + dt.diena

end;

Toliau ieškosime pirmosios metų savaitės trukmės. Pasinaudosime 2-ojo uždavinio funkcija naujieji_metai. Ji randa kurią savaitės dieną prasideda duotieji metai. Funkciją teks šiek tiek pakeisti (pritaikyti, kai data vaizduojama įrašu).

Pateikiame likusį algoritmo tekstą:

function savaitės_numeris (dt: data): integer;

{ nustato kelintai metų savaitei priklauso duotoji data }
  var trukmė: integer;

  function kel (m: metai): boolean;

    { ar metai m keliamieji }
  begin
    kel := (m mod 400 = 0) or
           (m mod 100 <> 0) and
           (m mod 4 = 0)

  end;

  function naujieji_metai (m: metai): integer;

  { randa pirmąją metų savaitės dieną }

  { savaitės dienos numeruojamos: }

  {1 – pirmadienis, 2 – antradienis, ..., 7 – sekmadienis }
    var mt: metai; { pasirenkami metai, kurių pirmoji diena sekmadienis }
        d, i: integer;

  begin
    mt := 1978; { pasirenkami metai }
    { kurių pirmoji savaitės diena yra sekmadienis}
    d := 7;

    for i := mt-1 downto m do { jei duoti metai yra ankstesni }
        if kel (i)

           then d := ((d - 2) + 7) mod 7

           else d := ((d - 1) + 7) mod 7;

    for i := mt to m-1 do  { jei duoti metai yra vėlesni }
        if kel (i)

           then d := (d + 2) mod 7

           else d := (d + 1) mod 7;

    if d = 0

       then d := 7;

    naujieji_metai := d

  end;

begin
  trukmė := 7 - naujieji_metai (dt.mt) + 1;

  savaitės_numeris := (d_sk(dt) - trukmė - 1) div 7 + 2

end;

	Testo nr.
	Pradiniai duomenys
	Rezultatas
	Paaiškinimai

	1
	1978 01 02
	 2
	Metai prasideda sekmadieniu

	2
	1969 03 02
	 9
	Metai paprastieji, prasideda trečiadieniu

	3
	1967 03 05
	10
	Metai paprastieji, prasideda sekmadieniu

	4
	1972 03 01
	10
	Metai keliamieji, prasideda šeštadieniu

	5
	1998 12 27
	52
	Metai prasideda ketvirtadieniu;  

	6
	1998 12 28
	53
	12.27 – sekmadienis

	7
	2000 06 31
	27
	Atsitiktinis testas


17. Lygybė. Susitarkime, kad langeliai numeruojami iš viršaus į apačią ir iš kairės į dešinę. Šiame uždavinyje reikia paprasčiausiai patikrinti visus galimus variantus – t. y. perrinkti visus keturženklius skaičius. Jų nėra daug. 

Tačiau tikrinti reikia ne visai aklai. Jei imsime keturženklį skaičių 6789 ir jo skaitmenų sumą kelsime ketvirtuoju laipsniu, gausime skaičių daug didesnį už maxint. Todėl perrinksime tik tuos keturženklius skaičius, kurių skaitmenų suma, pakelta ketvirtuoju laipsniu, yra keturženklis skaičius – t. y. skaičius, kurių skaitmenų suma yra mažesnė už 10.

Sąlygoje pasakyta, kad vienodos raidės žymi tuos pačius skaitmenis, tačiau nepasakyta, ar skirtingos raidės žymi skirtingus skaitmenis. Todėl nebūtinai visi ieškomo skaičiaus skaitmenys turi būti skirtingi.

program lygybė;

  var a, b, c, d: integer;

begin
  for a := 1 to 9 do
    for b := 0 to 9 do
      for c := 0 to 9 do
        for d := 0 to 9 do
          if a + b + c + d < 10

          then if sqr (sqr (a + b + c + d)) =

                  a * 1000 + b * 100 + c * 10 + d

                  then writeln (a*1000+b*100+c*10+d)

end.

	Rezultatas
	Paaiškinimas

	2401
	(2 + 4 + 0 + 1)4 ( 2401


18. Paprastas kelias. Prieš sudarydami šio uždavinio algoritmą pirmiausia šiek tiek pamastykime. Kaip eitume tarp dviejų lentos langelių? Jei eitume tik langeliais, kurie liečiasi kraštinėmis (t.y. eitume tik horizontaliai ar vertikaliai), tai trumpiausias kelias būtų dviejų atkarpų laužtė (atskiru atveju – viena, kai langeliai A ir B toje pačioje tiesėje). 

Sąlygoje leidžiama eiti ir įstrižai (t. y. langeliais, kurių kampai liečiasi). Tuomet ankstesnis kelias nebus trumpiausias. Reikia prisiminti akivaizdžią taisyklę: kelias įžambine visuomet trumpesnis negu statiniais. 

Taisyklę taikome kombinuotai: einame įžambine, kol susilyginame su langeliu B, likusią kelio dalį einame horizontaliai arba vertikaliai 

Visiškai nesvarbu, kur yra langelis B: prieš langelį A, po jo, virš ar žemiau jo. Visais atvejais elgiamės analogiškai: pirmiausia einame įstrižai, po to – tiesiai į B

Vadinasi, pirmiausia reikia rasti langelių A ir B koordinačių skirtumus (absoliutiniu dydžiu):

dx = (xa - xb(
dy = (ya - yb(
Mažesnysis iš dydžių dx ir dy rodo, kiek langelių reikės eiti įstrižai. Pastebėsime, kad kai langeliai A ir B yra viename stulpelyje arba vienoje eilutėje, tuomet vienas iš skaičių dx arba dy lygus 0, o tai reiškia, kad įstrižai nereikės eiti. 

Paėjus įstrižai, atsiduriama langelyje, kuris yra vienoje tiesėje su langeliu B. Dabar iš didesniojo iš dydžių dx ir dy atėmus mažesnįjį sužinosime, kiek langelių teks eiti tiese. 

Taigi trumpiausias kelias lygus:

min (dx, dy) + max (dx, dy) - min (dx, dy) + 1 = max (dx, dy) + 1

Vienetas pridedamas todėl, kad langelis A taip pat priklauso ieškomam keliui.

type langelis = record
                  x, y: 1..100

                end;

function kelias (A, B: langelis): integer;

  var dx, dy: integer;

begin
  dx := abs (A.x - B.x);

  dy := abs (A.y - B.y);

  if dx > dy

     then kelias := dx + 1

     else kelias := dy + 1

end;

	Testo nr.
	Pradiniai duomenys
	Rezultatas
	Paaiškinimai

	1
	10 11 9 11
	 2
	A ir B gretimi langeliai

	2
	85 46 85 29
	18
	A ir B horizontalioje tiesėje

	3
	46 98 12 98
	35
	A ir B vertikalioje tiesėje

	4
	2 9 65 72
	64
	A ir B vienoje įstrižainėje

	5
	64 32 14 35
	51
	A yra žemiau ir kairiau langelio B

	6
	87 45 26 21
	62
	A yra žemiau ir dešiniau langelio B

	7
	7 2 96 19
	90
	A yra aukščiau ir kairiau langelio B

	8
	10 33 15 12
	22
	A yra aukščiau ir dešiniau langelio B

	9
	4 99 100 1
	99
	A ir B yra priešinguose lentos kampuose


19. Lenta su skylėmis. Populiariausia šio uždavinio sprendimo idėja būna tokia: iš eilės imame visus lentos kvadratėlius ir bandome juos plėsti žemyn, į dešinę, stengdamiesi sudaryti kuo didesnį kvadratą be skylių. Taip patikrinę visus kvadratus, rasime didžiausiąjį.

Pasitelkus dinaminio programavimo metodą galima parašyti daug trumpesnį ir efektyvesnį algoritmą. Nagrinėkime sudarinėjamų kvadratų apatinius kairiuosius kampus. Sudarykime lentelę, turinčią tiek elementų, kiek lentoje yra langelių:

c[i, j], (1 ( i ( n, 1 ( j ( m),  

čia c[i, j] reikšmė būtų lygi didžiausio kvadrato be skylių, kurio apatinis dešinysis kampas yra langelyje (i, j), ilgiui. 

Jei langelyje (i, j) yra skylė, akivaizdu, kad

c[i, j] ( 0.

Jei šis langelis yra viršutinėje eilutėje ar pirmame stulpelyje tai

c[i, j] ( 1.

Kitais atvejais c[i, j] reikšmė priklauso nuo to, kiek „erdvės“ yra nuo šio langelio į kairę, į viršų bei įstrižai. Tai galime sužinoti patikrinę kairėje, įstrižai ir viršuje esantiems langeliams atitinkančias lentelės c reikšmes. Iš trijų reikšmių išrinkę mažiausią ir ją padidinę vienetu, gausime ieškomąją c[i, j] reikšmę. 

Tai galima užrašyti formule:

                    0, jei langelyje [i, j] yra skylė;

c[i, j] (         1, jei i ( 1 arba j ( 1 ir langelyje skylės nėra;

                    min (c[i - 1, j], c[i, j - 1], c[i - 1, j - 1]) + 1, jei i > 1, j > 1 ir

                                                                                       langelyje skylės nėra.

Vieną kartą peržiūrėję pradinę lentą, užpildome lentelę c. Sąlygoje pateiktam pavyzdžiui užpildyta lentelė atrodytų taip:

	1
	1
	1
	1
	1
	1
	1
	1
	1

	1
	2
	2
	2
	0
	1
	2
	2
	2

	1
	0
	1
	2
	1
	1
	2
	3
	3

	1
	1
	1
	2
	2
	2
	2
	3
	4

	1
	2
	2
	2
	3
	3
	0
	1
	2

	1
	2
	3
	3
	3
	4
	1
	1
	2


Norint sužinoti kvadrato, turinčio ilgiausią briauną, kraštinės ilgį, reikia peržiūrėti lentelę c ir išrinkti didžiausią reikšmę.

Beje, skaičiuojant i-osios eilutės elementų reikšmes naudojamasi jau apskaičiuotomis 
i-osios ir (i-1)-osios eilučių reikšmėmis. Taigi neverta įsiminti visos lentelės: užtenka dviejų eilučių – dabar pildomos ir prieš tai buvusios. O didžiausią reikšmę galime išrinkti iš karto pildydami lentelę.

const MAX = 100; { maksimalus lentos dydis }
type lenta = array [1..MAX, 1..MAX] of (kvadr, skylė);

function min (a, b, c: integer): integer;

{ randa mažiausią iš trijų skaičių }
  var mn: integer;

begin
  mn := a;

  if mn > b

     then mn := b;

  if mn > c

     then mn := c;

  min := mn

end;

function kvadratas (lent: lenta;

                    m, n: integer): integer;

  var did, i, j: integer;

      c1, c2: array [1..MAX] of integer; { viena lentelės c eilutė }
begin
  did := 0; { didžiausio kvadrato kraštinė }
  { užpildome pirmąją lentelės c eilutę }
  for j := 1 to n do
    if lent [1, j] = skylė

       then c1[j] := 0

       else begin
              c1[j] := 1;

              if c1[j] > did

                 then did := c1[j]

            end;

  { pildome kitas lentelės c eilutes }
  for i := 2 to m do
    begin
      for j := 1 to n do
        begin
          if lent[i, j] = skylė then c2[j] := 0

          else 

            if j = 1 then c2[j] := 1

            else c2[j] := min (c2[j-1], c1[j-1], c1[j]) + 1;

          if did < c2[j]

             then did := c2[j]

        end;

      c1 := c2

    end;

  kvadratas := did

end;

	Testo nr.
	Lentos matmenys ir skylių skaičius lentoje
	Rezultatas
	Paaiškinimai

	1
	100 100

7
	59
	Nedaug skylių, tačiau jos išblaškytos po visą lentą

	2
	10 11

3
	6
	Nedidelė lenta

	3
	100 100

120
	39
	Didelis atsitiktinis testas

	4
	100 100

99
	90
	Visos skylės yra toje pačioje eilėje

	5
	100 100

169
	90
	Didelis atsitiktinis testas


Papildomi uždaviniai (per spaudą)

20. Faktorialas ir trejeto laipsnis. Šis uždavinys iš pirmo žvilgsnio atrodo labai lengvas: reikia apskaičiuoti duotojo skaičiaus faktorialą ir dalyti jį iš trijų tol, kol dalijasi be liekanos. Tačiau faktorialo reikšmė labai greitai didėja, todėl galima suskaičiuoti tik nedidelių pradinių duomenų faktorialus. Skaičiuojant faktorialo reikšmę didesniems duomenims, greitai įvyksta perpildymas ir gaunami netikslūs rezultatai. Todėl reikėtų tirti faktorialą sudarančius dauginamuosius:

n! = 1 ( 2 ( 3 ( 4 ( … ( n

Radę, iš kurių trejeto laipsnių dalijasi kiekvienas dauginamasis, ir tuos laipsnius sudėję, gausime ieškomąjį rezultatą. Galime pastebėti, kad nebūtina tirti kiekvieną dauginamąjį: iš 3 dalysis tik kas trečias sandaugos elementas.

Yra dar geresnis sprendimas. Faktorialą galima užrašyti tokiu būdu:

n! = 1 ( 2 ( 3 ( 4 ( 5 ( 6 ( 7 ( 8 ( 9 ( 10 (... ( n;  

3 ( 1            3 ( 2          3 ( 3...     3 ( (n div 3)  

Matome, kad kas trečias dauginamasis dalosi iš trijų, iš šių skaičių kas trečias dar kartą dalysis iš trijų ir t. t. Taigi duotąjį skaičių padaliję iš 3, gautą dalmenį dar kartą padaliję iš trijų ir t. t. bei gautus dalmenis sudėję, gausime ieškomąjį rezultatą.

function laipsnis (n: integer): integer;

  var laips: integer;

begin
  laips := 0;

  while n > 2 do
    begin
      n := n div 3;

      laips := laips + n

    end;

  laipsnis := laips

end;

	Testo nr.
	Pradinis duomuo
	Rezultatas
	Paaiškinimai

	1
	2
	0
	Ribinis atvejis

	2
	7
	2
	n! < maxint

	3
	82
	40
	Trys atvejai, kuriais tikrinama, ar pastebima

	4
	83
	40
	kada padidėja rezultatas

	5
	84
	41
	

	6
	32767
	16376
	Ribinis atvejis – pradinis duomuo lygus maxint


21. Stačiakampis. Šiame uždavinyje daugiau matematikos, nei programavimo, be to, jį galima spręsti skirtingais metodais. 

Nors stačiakampį galima vienareikšmiškai nusakyti trijų kampų koordinatėmis, laikome, kad yra duotos visos keturios. Jos išvardytos pagal laikrodžio rodyklę.

Akivaizdu, kad jei stačiakampio kraštinės yra lygiagrečios koordinačių ašims, taškų kiekį stačiakampio viduje rasti labai paprasta 

Kai stačiakampis yra pasviręs koordinačių ašių atžvilgiu, rasti taškų kiekį yra sunkiau.

Papildykime duotąjį stačiakampį iki koordinačių ašims lygiagretaus stačiakampio 

Taškai, esantys trikampių AED ir BCG (jie yra lygūs) viduje nepriklauso stačiakampiui ABCD. Jų skaičius lygus skaičiui taškų, esančių stačiakampio BLCG viduje: 

tšk_sk = (y2 - y3 - 1) ( (x2 - x3 - 1)

Be to, reikia sužinoti, ar atkarpoje AD (taškus, esančius atkarpoje BC, jau atmetėme) yra ieškomų taškų (jie nėra stačiakampio ABCD viduje, todėl juos reikia pridėti prie stačiakampio BLCG vidinių taškų). 

Taškų, kuriuos kerta atkarpa AD, skaičius lygus atkarpų ED ir EA didžiausiam bendrajam dalikliui minus 1.

Analogiškai suskaičiuojamas ir trikampiuose AFB ir DHC esančių taškų skaičius: randamas taškų skaičius stačiakampio DHCM viduje ir papildomas taškų, esančių atkarpose AB ir CD, skaičiumi. 

Ieškomasis stačiakampio ABCD viduje esančių taškų skaičius bus gautas iš stačiakampio EFGH vidinių taškų skaičiaus atėmus stačiakampių BLCG ir DHCM taškus.

Šiame algoritme remiamasi teiginiu, kad stačiojo trikampio įžambinei priklausančių taškų su sveikosiomis koordinatėmis skaičius lygus abiejų statinių didžiausiam bendrajam dalikliui minus 1, jei neįskaitomi įžambinės galai.

Įrodysime šį teiginį. Imkime bet kokį statųjį trikampį, kurio statinių ilgiai yra sveikieji skaičiai. Nubrėžkime keletą trikampio pagrindui lygiagrečių atkarpų. Tose vietose, kur atkarpos kerta įžambinę c, nuleiskime aukštines į trikampio pagrindą. Taip gausime keletą prigludusių prie įžambinės mažų trikampių 

Visi mažieji trikampiai bus panašūs į didįjį trikampį (jų visi atitinkami kampai lygūs).

Dabar pakoreguokime tuos mažus trikampius: juos nupieškime taip, kad visi mažieji trikampiai būtų vienodi. Taip bus tik tada, jei trikampio vertikalųjį statinį padalysime į lygias dalis ir iš įstrižainę kertančių taškų nuleisime aukštines. Kadangi trikampiai yra panašūs, tai, jei sutampa vienas iš statinių, turi sutapti ir kitų kraštinių ilgiai 

Norint, kad trikampių vertikalių statinių ilgiai būtų sveikieji skaičiai, didžiojo trikampio statinį b reikia dalyti į k dalių, čia k – bet koks b daliklis. Antrasis mažųjų trikampių statinis bus sveikasis skaičius tik tada, jei k bus ir a daliklis. 

Abu kiekvieno mažojo trikampio įstrižainės galai priklauso įstrižainei c. Jei kiekvieno jų abiejų statinių ilgai sveikieji skaičiai, tai ir taškų, esančių abiejuose įstrižainės galuose, koordinatės bus sveikieji skaičiai. 

Įsitikinome, kad jei trikampį galima aukščiau paaiškintu būdu padalyti į mažesnius trikampius, kurių statinių ilgiai yra sveikieji skaičiai, tai mažesnių trikampių įstrižainių galai priklausys pradinio trikampio įstrižainei ir jų koordinatės bus sveikieji skaičiai. 

Kiek daugiausiai tokių taškų gali būti? Tiek, kiek yra mažųjų trikampių minus 1. Na o jų daugiausiai gali būti tiek, koks yra statinių a ir b didžiausias bendrasis daliklis.

Nors sprendimo idėjos aprašas ir įrodymas gana ilgi, pats algoritmas – trumpas:

type taškas = record
                x, y: integer;

              end;

function dbd (x, y: integer): integer;

{ randa skaičių x ir y didžiausiąjį bendrą daliklį }
begin
  if x = 0

     then dbd := y

     else dbd := dbd (y mod x, x)

end;  { dbd }
function max (x, y: integer): integer;

begin
  if x > y

     then max := x

     else max := y

end;

function taškai (a, b, c, d: taškas): integer;

{ randa stačiakampio viduje esančių taškų skaičių }
  var ilgis, plotis, { stačiakampio dydis }
      a1, b1, a2, b2, { trikampių statiniai }
      tašk, tašk1, tašk2: integer;

begin
  { randame gaubiančiojo stačiakampio dydį }
  plotis := max (abs (a.x - c.x), abs (b.x - d.x));

  ilgis := max (abs (a.y - c.y), abs (b.y - d.y));

  tašk := (plotis - 1) * (ilgis - 1);

  { rasime pirmo trikampio kraštines }
  a1 := abs (a.x - b.x);

  b1 := abs (a.y - b.y);

  tašk1 := (a1  - 1) * (b1 - 1) + dbd (a1, b1) - 1;

  { taip pat ir antro trikampio }
  a2 := abs (b.x - c.x);

  b2 := abs (b.y - c.y);

  tašk2 := (a2 - 1) * (b2 - 1) + dbd (a2, b2) - 1;

  taškai := tašk - tašk1 - tašk2

end;

	Testo nr.
	Pradiniai duomenys
	Rezultatas
	Paaiškinimai

	1
	0 0

0 15

16 15

16 0
	210
	Stačiakampio kraštinės lygiagrečios koordinačių ašims, jo apatinio kairiojo kampo koordinatės (0, 0)

	2
	6 8

6 45

38 45

38 8
	1116
	Stačiakampio kraštinės lygiagrečios koordinačių ašims

	3
	1 5

4 11

14 6

11 0
	68
	Stačiakampio kraštinės  nėra lygiagrečios koordinačių ašims, kiekvienai jo kraštinei priklauso lyginis taškų skaičius

	4
	0 1

1 11

11 10

10 0
	100
	Pradinis duomuo – kvadratas. Jo kraštinės nėra lygiagrečios koordinačių ašims, kraštinėms nepriklauso nė vienas taškas

	5
	0 5

1 8

13 4

12 1
	36
	Stačiakampio kraštinės nėra lygiagrečios koordinačių ašims, kiekvienai jo kraštinei priklauso nelyginis taškų skaičius


22. Lošimas degtukais. Bandysime sudaryti žaidimo strategiją.
Norėdami laimėti, turime stengtis sudaryti priešininkui tokią situaciją, kad abiejose krūvelėse būtų po lygiai degtukų arba jų skirtumas būtų trijų kartotinis. Pirmuoju atveju laimėsime, imdami po tiek pat degtukų, kiek jų ima priešininkas. Antruoju atveju turime imti po tiek degtukų, kad po mūsų ėjimo krūvelėse esančių degtukų skirtumas būtų dalus iš trijų; taip darome tol, kol liks viena krūvelė. Šioje krūvelėje esančių degtukų skaičius turi būti dalus iš trijų. Tada laimėjimo strategija dar paprastesnė: jei priešininkas paims vieną degtuką, mums reikia imti du, o jei priešininkas paims du – turime imti vieną. Tokiu būdu mums atiteks paskutiniai degtukai. Vadinasi, jei krūvelės nelygios ir degtukų skirtumas nedalus iš trijų, imsime iš didesniosios.

Jeigu krūvelėse jau iš karto duota po lygiai degtukų arba degtukų skirtumas dalus iš trijų – neturime vilčių laimėti. Tokiu atveju galima elgtis įvairiai: pranešti apie pralaimėjimą arba imti vieną degtuką iš didesnės krūvelės (kad atitolintumėme pralaimėjimą ar galbūt sulauktumėme varžovo klaidos).

procedure degtukai (k1, k2: integer; { degtukų skaičius krūvelėse }
                    var imti: integer; { kiek degtukų imti }
                    var kr_nr: integer); { iš kurios krūvelės imti }
begin
  if k2 > k1

     then kr_nr := 2

     else kr_nr := 1;

  imti := abs (k1 - k2) mod 3;

  if imti = 0            { vis tiek pralošta, }
     then imti := 1      { tai imame mažiau degtukų }
end;

Nors sąlygoje ir neprašoma, tačiau šis algoritmas turėtų pranešti, išlošiama ar pralošiama. Taip būtų patogiau.

	Testo nr.
	Pradiniai duomenys
	Rezultatai
	Paaiškinimai

	1
	   1    1
	1 1
	Paprastas atvejis; egzistuoja vienintelis galimas ėjimas

	2
	   5    7
	2 2
	Paprastas atvejis, kai išlošiama

	3
	  49   48
	1 1
	Išlošiama sulyginat krūveles

	4
	1069 1070
	1 2
	

	5
	4025   33
	2 1
	Išlošiama, nes po ėjimo degtukų skirtumas tampa 

	6
	  88    3
	1 1
	lygus trijų kartotiniui

	7
	  48   48
	1 1
	Išlošti neįmanoma, nes abi krūvelės lygios

	8
	   3   87
	1 2
	Išlošti neįmanoma, nes degtukų skirtumas tarp krūvelių yra trejeto kartotinis


Aukštesniosioms mokykloms

23. Dvi progresijos. Nesunku pastebėti, kad iš abiejose progresijose esančių vienodų elementų galima sudaryti naują aritmetinę progresiją, kurios skirtumas – duotųjų aritmetinių progresijų skirtumų mažiausias bendrasis kartotinis, o pirmasis narys – pirmasis sutampantis abiejų progresijų narys. 

Tuo nesunku įsitikinti. Duotųjų progresijų pirmasis sutampantis narys yra 21; antrasis pirmosios progresijos sutampantis narys yra 21 + 4k (nes jos skirtumas – 4), o antrosios – 21 + 5m (nes jos skirtumas – 5), vadinasi 4 ( k = 5 ( m. Akivaizdu, kad k = 5, m = 4, nes 4 ir 5 didžiausias bendrasis daliklis yra 1. Tokiu būdu gauname naują aritmetinę progresiją, kurios pirmasis narys yra 21, o skirtumas lygus 20.

Ieškoma suma lygi naujos progresijos pirmųjų n narių sumai. Ji gaunama pagal formulę:

           2p1 + d(n - 1)

S (                        n
                      2

čia p1 – pirmasis progresijos narys, d – skirtumas.

Rašydami algoritmą pasinaudosime dviem funkcijomis, kurios padės surasti mažiausią bendrąjį kartotinį:

function dbd (x, y: integer): integer;

{ didžiausias bendrasis daliklis }
begin
  if x = 0

     then dbd := y

     else dbd := dbd (y mod x, x)

end;

function  mbk (x, y: integer): integer;

{ mažiausias bendrasis kartotinis }
begin
  mbk := x div dbd (x, y) * y

end;

Pateikiame algoritmą, ieškantį naujos progresijos pirmųjų narių sumos:

function suma (n: integer): integer;

  var a,           { pirmos progresijos pirmasis narys }
      da, db,      { bei abiejų progresijų skirtumai }
      narys,       { naujos progresijos pirmasis narys }
      d: integer; { naujos progresijos skirtumas }
begin
  a := 17; da := 4;  { pirmoji progresija }
  db := 5;           { antroji progresija }
  d := mbk (da, db);

  narys := a + da;   { žinome, kad sutampa antrieji progresijų nariai }
  suma := (2 * narys + d * (n - 1)) div 2 * n

end;

	Testo nr.
	Pradinis duomuo
	Rezultatas

	1
	1
	21

	2
	2
	62

	3
	10
	1110

	4
	28
	8148

	5
	37
	14097

	6
	56
	31976


Paskutinis testas pateikia didžiausią galimą n, kai suma dar neviršija maxint.
24. Langelių tušavimas. Norint išspręsti šį uždavinį, reikia išnagrinėti visas galimas n reikšmes.

Iš karto galime pastebėti, kad negalėsime užtušuoti langelių, kai n ( 1 (tada gretimų užtušuotų langelių skaičius yra 0), n ( 2 (gretimų užtušuotų langelių skaičius yra 1, t. y. nelyginis skaičius), n ( 3 (2 langeliai turės tik po vieną užtušuotą kaimyną).

Kai n = 4, tai langelius galime užtušuoti. Kai n = 5 ir n ( 6 – užtušuoti negalime. Kai n = 7, langelius užtušuoti galime 

Kai n – lyginis skaičius ir nemažesnis už 8, langelius užtušuoti irgi galime 

Belieka ištirti, ar galime užtušuoti n langelių, kai n – nelyginis skaičius. Nesunku pastebėti, kad jei turime užtušuotą lyginį skaičių langelių, tai užtušavę tris papildomus langelius aplink kampinį stačiakampio langelį, gausime nelyginį užtušuotų langelių skaičių 

Tokiu būdu galime gauti visus nelyginius skaičius didesnius už 5, išskyrus 9.

Taigi, jei n reikšmė lygi 1, 2, 3, 5, 6, 9, tai užtušuoti negalime, visoms kitoms n reikšmėms užtušuoti langelius galime. 

Gali būti ir kitokie langelių išdėstymo variantai: jei n - 1 dalus iš 3.

program langelių_tušavimas;

  var n, eil, i: integer;

      lyg: boolean; { ar langelių skaičius lyginis }
begin
  write ('Įveskite langelių skaičių: ');

  readln (n);

  case n of
     1, 2, 3, 5, 6, 9: writeln ('UŽTUŠUOTI NEGALIMA');

     4: begin
              writeln ('**');

              writeln ('**')

        end;

     7: begin
              writeln ('** ');

              writeln ('***');

              writeln (' **')

        end;

     else  begin
              writeln ('***');

              lyg := n mod 2 = 0;

              if lyg then eil := (n - 6) div 2

                     else eil := (n - 9) div 2;

              for i := 1 to eil do
                writeln ('* *');

              write ('***');

              if not lyg

                 then begin
                        writeln ('*');

                        writeln ('  **')

                      end
            end
  end
end.

	Testo nr.
	Pradinis duomuo
	Rezultatas
	Paaiškinimai

	1
	1
	UŽTUŠUOTI NEGALIMA
	

	2
	2
	UŽTUŠUOTI NEGALIMA
	

	3
	3
	UŽTUŠUOTI NEGALIMA
	Visi atvejai, kai užtušuoti negalima

	4
	5
	UŽTUŠUOTI NEGALIMA
	

	5
	6
	UŽTUŠUOTI NEGALIMA
	

	6
	9
	UŽTUŠUOTI NEGALIMA
	

	7
	4
	**

**
	Du atskiri atvejai, kai užtušuoti galima

	8
	7
	**

***

 **
	

	9
	12
	...
	n - lyginis

	10
	15
	...
	n - nelyginis

	11
	23
	...
	n - nelyginis;


25. Daugiakampis. Šis uždavinys sutampa su 29 uždaviniu: ten ieškokite jo sprendimo.

II etapas

26. Atsiskaitymas doleriais. Turime m dolerių. Žmogaus indėlis į darbą yra p%. Tuomet jis turi gauti (p ( m)/100 dolerių. Kiekvienam žmogui nesunkiai galime apskaičiuoti jo tikrą užmokestį. 

Tarkime, kad žmogus uždirbo 2,35 dolerio. Jam galime duoti arba 2 arba 3 dolerius. Vienu atveju paklaida būtų 0,35 cento, kitu – 0,65 cento. Taigi šiam žmogui reikėtų duoti 2 dolerius. Situacija šiek tiek pasikeičia, kai uždarbį reikia pasidalyti keliems žmonėms. 

Pirmiausia kiekvienam žmogui išdalinama po sveikąjį uždirbtų dolerių skaičių (t. y. jei žmogus uždirbo 6,80 dolerių, jam bus išmokėta 6 doleriai). Sakykime, išdalijus liko n dolerių.

Tuomet žmones vertėtų surikiuoti mažėjimo tvarka pagal jų uždirbtus centus (t. y. nekreipiant dėmesio į tai, kiek sveikų dolerių uždirbta, nes juos žmonės jau atsiėmė). Ir pirmiesiems n žmonių išdalijama dar po 1 dolerį.

Panagrinėkime sąlygoje pateiktą pavyzdį. Yra trys žmonės. Pirmasis jų uždirbo 32,67 dolerio, antrasis – 59,4 dolerio, trečiasis – 6,93 dolerio. Jiems iš karto išmokama 32 + 59 + 6 ( 97 doleriai. Lieka 2 doleriai. Daugiausia centų uždirbo trečiasis ir pirmasis žmonės. Jiems duodama dar po vieną dolerį. Taigi išmokama 33, 59 ir 7 doleriai. Paklaidų suma lygi: 

(33 - 32,67) + (59,4 - 59) + (7 - 6,93) ( 0,8.

const MAXN = 100; { maksimalus žmonių skaičius }
type lentelė = array [1..MAXN] of integer;

     likutis = record
                 suma: real;

                 nm: integer { kuriam žmogui priklauso likutis }
               end;

     likučiai = array [1..MAXN] of likutis;

procedure rikiuoti (n: integer;

                    var cent: likučiai);

{ surikiuoja skaičius nedidėjimo tvarka }
{ galima naudoti bet kurį rikiavimo algoritmą }

{ programoje panaudotas burbulo metodas, algoritmas paimtas iš 52 užd. }

begin
  ...

end;

procedure padalyti (n, { žmonių skaičius }
                    m: integer; { dolerių skaičius }
                    p: lentele; { indėlis procentais }
                    var sum: lentelė { išdalyti pinigai });

  var i, liko: integer;

      užd: real;

      cent: likučiai;

begin
  liko := m;

  { išdalysime sveikąją dalį ir suskaičiuosime uždirbtus centus }
  for i := 1 to n do
    begin
      sum[i] := p [i] * m div 100;

      liko := liko - sum[i];

      cent[i].suma := p [i] * m / 100  - sum[i];

      cent[i].nm := i

    end;

  { surikiuojama pagal uždirbtus centus }
  rikiuoti (n, cent);

  { išdalijami likę doleriai }
  for i := 1 to liko do
    sum[cent[i].nm] := sum [cent[i].nm] + 1

end;

	Testo nr.
	Pradiniai duomenys
	Rezultatai
	Paaiškinimai

	1
	1500 8

20 17 13 10 5 15 14 6
	300 255 195 150 75 225 210 90
	Užmokestis išsidalija tiksliai; paklaidų suma – 0

	2
	55 5

60 15 10 10 5
	33 8 6 5 3
	Dviem žmonėms, kurių indėlis vienodas, mokama skirtingai. Paklaidų suma – 1.5

	3
	45 7

12 14 18 10 21 23 2
	5 6 8 5 10 10 1
	Atsitiktinis testas, paklaidų suma – 2.3

	4
	26 5

40 1 20 14 25
	10 0 5 4 7
	Vienas žmogus visai negauna pinigų, paklaidų suma – 1.72


27. Tiesės. Pirmiausia suskaičiuokime visus tiesių tarpusavio susikirtimo taškus. Jei kurios nors dvi tiesės sutampa, laikome, kad jos neturi bendrų taškų. Norint sudaryti keturkampį, susikirtimo taškų skaičius turi būti ne mažesnis kaip keturi. Jei kiekviena tiesė kertasi su visomis kitomis skirtinguose taškuose, gali būti daugiausia šeši susikirtimo taškai.

Panagrinėkime šiuos tiesių susikirtimo variantus. 

1. Tiesės kertasi keturiuose taškuose. Taip gali atsitikti, jei 

a) tiesės yra poromis lygiagrečios;

b) trys tiesės kertasi viename taške, ir nėra lygiagrečių tiesių.

Pirmuoju atveju visada gausime keturkampį, antruoju atveju jo niekada negausime, nes trys iš keturių taškų bus vienoje tiesėje.

2. Tiesės kertasi penkiuose taškuose. Taip gali atsitikti tik tada, kai dvi tiesės yra lygiagrečios, o likusios dvi kertasi:

Matome, kad pirmuoju atveju keturkampį sudaryti galima, antruoju – ne.

3. Tiesės kertasi šešiuose taškuose. Taip yra, kai jokios trys tiesės nesikerta viename taške ir jokios dvi tiesės nėra lygiagrečios:

Šiuo atveju visuomet galima sudaryti keturkampį. 

Dabar nesunkiai galime nustatyti, ar įmanoma sudaryti keturkampį. Jei yra šeši susikirtimo taškai, tai tiesės sudaro keturkampį. Jei penki – reikia patikrinti, ar dviejų nelygiagrečių tiesių susikirtimo taškas yra lygiagrečių tiesių (išorėje”, o jei keturi – pakanka patikrinti, ar yra lygiagrečių tiesių.

Jei žinome, kad tiesės sudaro keturkampį, reikia rasti koordinates. Kai susikirtimo taškai keturi, viskas paprasta – tie taškai yra ieškomieji. Kai yra penki susikirtimo taškai, reikia atmesti nelygiagrečių tiesių susikirtimo tašką, o likę bus keturkampio koordinatės. 

Kebliau, kai yra 6 susikirtimo taškai. Galima ieškoti tokiu būdu. Kiekviena tiesė kertasi su kitomis trimis. Vienas susikirtimo taškas ant konkrečios tiesės yra tarp kitų dviejų (27.3 pav. žr. 4,5 ir 6 taškus). Iš šių trijų taškų tiesei A priklauso taškai 4 ir 5, tiesei B – 5 ir 6, tiesei 
C – 4, tiesei D – 6. Taigi paskutinysis taškas bus tiesių C ir D susikirtimo taškas.

Gali kilti klausimas, kaip nustatyti, ar tiesės yra lygiagrečios. Tarkime, kad duotos dvi tiesės y ( a1x + b1, y ( a2x + b2. Jos yra lygiagrečios, kai a1 ( a2.

Jei dvi tiesės kertasi, jų susikirtimo taško (x, y) koordinatės apskaičiuojamos pagal formulę:

x ( (b2 - b1) / (a1 - a2);

y ( (a1b2 - a2b1) / (a1 - a2)

const MAX = 6; { maksimalus susikirtimų skaičius }
      N = 4;   { maksimalus tiesių skaičius }
type tiesė = record
               a, b: integer

             end;

     tiesės = array [1..N] of tiesė;

     susikirt = array [1..MAX] of 

                  record
                    nr1, nr2: integer; { kurios tieses kertasi }
                    x, y: real;        { koordinatės }
                    prikl: boolean     { ar priklauso keturkampiui }
                  end;

      aibė = set of 1..4;

function lyg (t1, t2: tiese): boolean;

{ nustato, ar dvi tiesės yra lygiagrečios }
begin
  lyg := t1.a = t2.a

end; { lyg }
procedure kertasi (t: tiesės;       { tiesės }
                   i, j: integer;   { jų numeriai }
                   var s: susikirt; { susikirtimo taškų lentelė }
                   var sk: integer  { jų skaičius });

{ randa tiesių i ir j susikirtimo tašką }
  const eps = 0.00000001;

  var buvo: boolean;

begin
  if t[i].a <> t[j].a then { jei tiesės nelygigagrečios }
   begin
     sk := sk + 1;

     s[sk].nr1 := i;

     s[sk].nr2 := j;

     s[sk].prikl := false;

     s[sk].x := (t[j].b - t[i].b) / (t[i].a - t[j].a);

     s[sk].y := (t[i].a*t[j].b - t[j].a*t[i].b) /

                (t[i].a - t[j].a);

     { tikrinsime, ar tokio taško nebuvo }
     buvo := false;

     for i := 1 to sk-1 do
       if (abs (s[i].x - s[sk].x) < eps) and
          (abs (s[i].y - s[sk].y) < eps)

          then buvo := true;

            if buvo then
               sk := sk - 1

   end
end; { kertasi }
function vidurinis (s: susikirt; a, b, c: integer): boolean;

{ patikrina, ar taškas b yra tarp taškų a ir c, kai }
{ žinoma, kad tie taškai tikrai yra vienoje tiesėje }
begin
  vidurinis := 

   (s[a].x <= s[b].x) and (s[b].x <= s[c].x) and 

   (s[a].y <= s[b].y) and (s[b].y <= s[c].y) or
   (s[c].x <= s[b].x) and (s[b].x <= s[a].x) and 

   (s[c].y <= s[b].y) and (s[b].y <= s[a].y)

end; { vidurinis }
function v_tiesė (s: susikirt; i, j: integer): boolean;

{ patikrina, ar susikirtimo taškai i ir j yra vienoje tiesėje }
begin
  v_tiesė := (s[i].nr1 = s[j].nr1) or (s[i].nr1 = s[j].nr2) or
             (s[i].nr2 = s[j].nr1) or (s[i].nr2 = s[j].nr2)

end; { v_tiesė }
procedure penki (t: tiesės;

                 sk: integer;     { susikirtimų skaičius }
                 var s: susikirt; { susikirtimo taškai }
                 var sudaro: boolean);

{ randa keturkampio taškus, kai tiesės kertasi penkiuose taškuose }
  var k: array [1..N] of integer; { kiek taškų kerta i-ąją tiesę }
      i, j, nm: integer;

      vid: boolean;

begin
  { į keturkampį įtraukiame lygiagrečių tiesių taškus }
  for i := 1 to N do
    k[i] := 0;

  for i := 1 to sk do
    begin
      k[s[i].nr1] := k[s[i].nr1] + 1;

      k[s[i].nr2] := k[s[i].nr2] + 1;

    end;

  for i := 1 to sk do
    if k[s[i].nr1] = 2

       then s[i].prikl := true

       else nm := i; { penktojo taško numeris }
  { tikriname, ar tinka penktasis, t. y. ar jis nėra viduryje tarp kurių nors dviejų taškų }
  vid := false;

  for i := 1 to sk do
   for j := i+1 to sk do
     if (i <> nm) and (j <> nm) and
        v_tiesė (s, i, j) and v_tiesė (s, i, nm)

        { jei visi trys taškai yra vienoje tiesėje }
        and vidurinis (s, i, nm, j)

        then vid := true;

  sudaro := not vid

end; { penki }
procedure vid (var s: susikirt; sk, i: integer;

               var aib: aibė);

{ randa i-osios tiesės vidurinįjį susikirtimo tašką }
{ ir jį įtraukia į keturkampį }
  var num: array [1..3] of integer;

      k, j, nm: integer;

begin
  k := 1;  j := 0;

  while j < 3 do { surandame visus tris i-osios tiesės susikirtimo }
    begin        { taškus }
      if (s[k].nr1 = i) or (s[k].nr2 = i)

         then begin
                j := j + 1;

                num[j] := k

              end;

      k := k + 1;

    end;

  if vidurinis (s, num[1], num[2], num[3])

     then nm := num[2]

     else if vidurinis (s, num[2], num[1], num[3])

             then nm := num[1]

             else nm := num[3];

  { tašką įtrauksime į keturkampį }
  s[nm].prikl :=  true;

  if s[nm].nr1 in aib

     then aib := aib - [s[nm].nr1]

     else aib := aib + [s[nm].nr1];

  if s[nm].nr2 in aib

     then aib := aib - [s[nm].nr2]

     else aib := aib + [s[nm].nr2];

end; { vid }
procedure šeši (t: tiesės;

                sk: integer { jų skaičius };

                var s: susikirt { susikirtimo taškai });
{ randa keturkampio taškus, kai tiesės kertasi šešiuose taškuose }
  var aib: aibė;

      i: integer;

begin
  aib := [];

  { rasime tiesių vidurinius taškus ir juos įjungsime į keturkampį }
  for i :( 1 to N do

    vid (s, sk, i, aib);

  { rasime ketvirtą tašką }
  for i := 1 to sk do
    if not s[i].prikl and (s[i].nr1 in aib) and 

       (s[i].nr2 in aib)

       then s[i].prikl := true

end; { šeši }

procedure keturkampis (t: tiesės; { duotosios tiesės }
                      var s: susikirt;{ keturkampio viršūnių koordinatės }
                      var sudaro: boolean; { ar sudaro keturkampį }
                      var sk: integer { susikirtimo taškų skaičius });

  var i, j: integer;

begin
  { rasime visus tiesių tarpusavio susikirtimo taškus }
  sk := 0;

  for i := 1 to N do
    for j := i + 1 to N do
      kertasi (t, i, j, s, sk);

  case sk of
    0, 1, 2, 3: sudaro := false;

    4: begin
          sudaro := lyg (t[1], t[2]) or 

                    lyg (t[1], t[3]) or lyg (t[1], t[4]);

          if sudaro

             then for i := 1 to sk do
                    s[i].prikl := true

         end;

    5: penki (t, sk, s, sudaro);

    6: begin
          sudaro := true;

          šeši (t, sk, s)

        end
   end
end; { keturkampis }
	Testo nr.
	Pradiniai duomenys
	Rezultatai
	Paaiškinimai

	1
	-2  -5

-2   4

-2  -1

-2   5
	NE
	Visos tiesės yra lygiagrečios

	2
	-2  22

 1 -11

 0  10

 0  10
	NE
	Trys tiesės kertasi trijuose taškuose, ketvirta tiesė sutampa su pirma; Sutampančios tiesės yra horizontalios; Susikirtimo taškai: 1(3 tiesių (6, 10), 1(2 tiesių (11, 0), 2(3 (21, 10)

	3
	 2   5

 2   4

 3   1

 3   2
	4 13

3 11

3 10

2  8
	Keturi susikirtimo taškai, tiesės yra poromis lygiagrečios, susikirtimo taškai: 1(3 tiesių (4, 13), 1(4 tiesių (3, 11), 2(3 (3, 10), 2(4 (2, 8)

	4
	 2  -2

 1   3

-8  48

-2   6
	NE
	Keturi susikirtimo taškai, trys tiesės kertasi viename taške, susikirtimo taškai: 1(2(3 tiesių (5, 8), 1(4 tiesių (2, 2), 2(4 (1, 4), 3(4 (7, -8)

	5
	 2   1

 2  -9

-3  21

 0  -3
	 4  9

 6  3

-2 -3

 3 -3
	Penki susikirtimo taškai, sprendinys yra, susikirtimo taškai: 1(3 tiesių (4, 9), 2(3 tiesių (6, 3), 1(4 (-2, -3), 3(4 (8, -3), 2(4 (3, -3)

	6
	 2   1

 2  -9

-3  21

 7 -29
	NE
	Penki susikirtimo taškai, sprendinio nėra, susikirtimo taškai: 1(3 tiesių (4, 9), 2(3 tiesių (6, 3), 1(4 (6, 13), 2(4 (4, -1), 3(4 (5, 6)

	7
	 2  -4

 1  -8

-2  16

-3  36
	-4 -12

 5   6

 8   0

20 -24
	Šeši susikirtimo taškais: 1(2 tiesių (-4, -12), 1(3 tiesių (5, 6), 1(4 (8, 12), 2(3 (8, 0), 2(4 (11, 3), 3(4 (20, -24)

	8
	-1   6

 0   2

 0   2

 2 -18
	 0   6

12   6

10   2

 4   2
	Penki susikirtimo taškai (27.4 pav.)


III etapas

28. Skaičių lentelės atstatymas. Trūkstamą lentelės elementą galima mėginti atstatyti pagal toje eilutėje ar stulpelyje esančius skaičius, t. y. iš eilutės ar stulpelio sumos atimti visus kitus elementus ir ieškoti tų elementų sumos. Taip galima atstatyti trūkstamą skaičių, jei eilutėje ar stulpelyje nėra daugiau nežinomų skaičių. Jei yra, reikia mėginti atstatyti tolesnį trūkstamą skaičių.
Vieną kartą patikrinę visas eilutes ir stulpelius, uždavinį retai kada išspręsime. Panagrinėkime tokį pavyzdį:

	222
	333
	–
	1
	–

	–
	–
	4
	–
	14

	224
	–
	–
	6
	1014


Patikrinę eilutes ir stulpelius pirmą kartą gausime:

	222
	333
	–
	1
	1000

	2
	–
	4
	5
	14

	224
	–
	–
	6
	1014


Matome, kad atstatyti ne visi skaičiai, nors akivaizdu, kad dar galima juos atstatyti. Patikrinę antrą kartą gauname:

	222
	333
	444
	1
	1000

	2
	3
	4
	5
	14

	224
	336
	448
	6
	1014


Taigi jei patikrinus lentelę nors vieną skaičių pavyko atstatyti, reikia viską kartoti iš pradžių. Taip darome, kol atstatome skaičius arba kol tokiu būdu nebegalime atstatyti skaičių. Jei pavyko atstatyti, būtina patikrinti, ar atstatytoje lentelėje nėra klaidų. Mat atstatinėjant galime nepamatyti, kad pradinėje lentelėje įrašyti nekorektiški skaičiai.

const max = 10; { maksimalus lentelės dydis }
type lentelė = array [1..max, 1..max] of integer;

     masyvas = array [1..max] of integer;

{ žemiau surašyti globalieji kintamieji }
var eil, st: integer; { eilučių ir stulpelių skaičius lentelėje }
      lent: lentelė; { pradinė lentelė }
procedure trūksta (var nutrinta: integer; { kiek nutrintų skaičių }
                   var e, s: masyvas);

{ suskaičiuojame kiek yra trūkstamų elementų kiekviename stulpelyje }
{ bei eilutėje ir kiek iš viso yra nutrintų skaičių }
  var i, j: integer;

begin
  nutrinta := 0;

  for i := 1 to eil do
    e[i] := 0;

  for i := 1 to st do
    s[i] := 0;

  for i := 1 to eil do
    for j := 1 to st do
      if lent[i, j] = -maxint

         then begin
                e[i] := e[i] + 1;

                s[j] := s[j] + 1;

                nutrinta := nutrinta + 1

              end
end; { trūksta }
function teisinga (atst: lentelė): boolean;

{ ar nėra klaidų atstatytoje lentelėje }
  var sum_eil, sum_st: masyvas; { atitinkamų eilučių ir stulpelių sumos }
      gerai: boolean;

      i, j: integer;

begin
  gerai := true;

  for i := 1 to eil do
    sum_eil[i] := 0;

  for i := 1 to st do
    sum_st[i] := 0;

  for i := 1 to eil do
    for j := 1 to st do
       begin
         if j = st  { jei susumavome eilutę }
            then gerai := gerai and (sum_eil[i] = atst[i, j])

            else sum_eil[i] := sum_eil[i] + atst[i, j];

         if i = eil { jei susumavome stulpelį }
            then gerai := gerai and (sum_st[j] = atst[i, j])

            else sum_st[j] := sum_st[j] + atst[i, j]

       end;

  teisinga := gerai

end; { teisinga }
procedure atstatyti (var atst: lentelė; { atstatyta lentelė }
                     var klaida, { ar yra lentelėje klaidų }
                     per_daug: boolean { ar per daug prarasta skaičių });

  var e, s: masyvas;

      { trūkstamų skaičių kiekis atitinkamoje eilutėje ar stulpelyje }
      nutrinta, i, j, k, suma: integer;

      atstatyta: boolean; { ar peržiūrėję lentelę atstatėme nors vieną skaičių }
begin
  { suskaičiuojame, kiek yra trūkstamų elementų kiekviename stulpelyje }
  { ir eilutėje ir kiek viso nutrinta skaičių }
  trūksta (nutrinta, e, s);

  atst := lent;

  { bandysime atstatyti skaičius }
  repeat
    atstatyta := false;

    for i := 1 to eil do
       for j := 1 to st do
         if (atst[i, j] = -maxint) then { jei skaičius nutrintas }
            if (e[i] = 1) or (s[j] = 1) then
              begin
                nutrinta := nutrinta - 1;

                atstatyta := true;

                suma := 0;

                if e[i] = 1 then
                   begin { jei eilutėje trūksta vieno skaičiaus }
                     for k := 1 to st-1 do
                       if k <> j then
                         suma := suma + atst[i, k];

                         if j = st then
                           atst[i, j] := suma

                         else atst[i, j] := atst[i, st] - suma

                   end
                else begin { jei stulpelyje trūksta vieno skaičiaus }
                       for k := 1 to eil - 1 do
                         if k <> i then
                           suma := suma + atst[k, j];

                         if i = eil then
                            atst[i, j] := suma

                         else atst[i, j] := atst[eil, j] - suma

                     end;

                e[i] := e[i] - 1;

                s[j] := s[j] - 1

              end;

  until not atstatyta or (nutrinta = 0);

  { tikrinama, ar nėra klaidų }
  if nutrinta <> 0

     then per_daug := true

     else begin
            per_daug := false;

            klaida := not teisinga (atst)

          end
end; { atstatyti }
	Testo nr.
	Rezultatai
	Paaiškinimai

	1
	Atstatyti galima
	Atstatyti lengva, trūksta tik trijų skaičių

	2
	Atstatyti neįmanoma

Neteisinga pradinė lentelė
	Klaidingas apatinės eilutės paskutinysis stulpelis

	3
	Atstatyti galima
	Atstatyti galima, peržiūrėjus lentelę vieną kartą

	4
	Atstatyti neįmanoma

Per daug prarastų skaičių
	Atstatyti neįmanoma, prarasta vienu skaičiumi per daug

	5
	Atstatyti galima
	Norint atstatyti teks lentelę paržiūrėti daug kartų

	6
	Atstatyti galima
	Testas, kai lentelės dydis – maksimalus


29. Daugiakampis. Sužinoti, ar galima iš duotų taškų sudaryti daugiakampį yra nesunku: tereikia patikrinti, ar taškų yra daugiau negu du ir ar ne visi taškai yra vienoje tiesėje. Tikrinant taškų priklausomybę vienai tiesei, pasirenkami bet kurie du taškai, parašoma tiesės per du taškus lygtis:

x - x1  (    y - y1
x1 - x2      y1 - y2

čia (x1, y1) yra pirmojo pasirinkto taško koordinatės, (x2, y2) – antrojo.

Į lygtį įstatomos visų likusių taškų koordinatės. Jei visi taškai tenkina šią tapatybę, jie yra vienoje tiesėje.

const MAXN = 100; { didžiausias galimas taškų skaičius }
type taškas = record
                 x, y: integer

              end;

     tdaugiakampis = array [0..MAXN] of taškas;

     { taškai numeruojami nuo 0 iki n }
function ant_tiesės (A, B, C: taškas): boolean;

{ ar taškas C yra vienoje tiesėje su taškais A ir B }
begin
   ant_tiesės := (C.y - B.y) * (B.x - A.x) =

                 (C.x - B.x) * (B.y - A.y)

end; { ant_tiesės }
function ne_tiesėje (d: tdaugiakampis; 

                     n: integer): boolean;

{ patikrina ar ne visi duoti taškai yra vienoje tiesėje }
  var i: integer;

      nėra: boolean;

begin
  nėra := true;

  i := 2;

  { tikrinama, ar yra bent vienas taškas nepriklausantis tiesei, einančiai per d[0] ir d[1] }
  while nėra and (i <= n) do
    begin
      nėra := ant_tiesės (d[i], d[0], d[1]);

      i := i + 1

    end;

  ne_tiesėje := not nėra;

end;  { ne_tiesėje }
Antroji uždavinio dalis – sunumeruoti taškus taip, kad jie sudarytų daugiakampį (jei galima) yra sunkesnė. Atrodytų, kad bet kokį daugiakampį sudaryti nėra sunku: sąlygoje nepasakyta, kad daugiakampis turėtų būti iškilusis. Vadinasi, daugiakampis gali būti ir neiškilusis, ir žvaigždinis (turėti susikertančių kontūrų) – apie tokius daugiakampius rašoma įvairiuose matematikos žinynuose. 

Atrodo, kad tiktų paprasčiausias algoritmas: surikiuoti taškus koordinatės x didėjimo tvarka, jei yra keli taškai su vienoda x koordinate – surikiuoti juos y koordinatės didėjimo tvarka. Taip surikiuotas masyvas rodytų jungiamų taškų tvarką. Paskutiniame masyvo elemente įrašytą tašką sujungtumėme su pirmuoju tašku ir turėtume gauti daugiakampį (kuris gali turėti ir susikertančių kontūrų ir didesnių nei 1800 kampų). Tačiau nesunku įsitikinti, kad taip jungdami taškus, galime gauti figūras, su įvairiomis „uodegomis“.

Jų negalima pavadinti daugiakampiais. Akivaizdu, kad minėtas algoritmas netinka. 

Taigi, uždavinio esmė yra surikiuoti duotus taškus taip, kad jungiant juos iš eilės gautos naujos atkarpos nors dalinai nesutaptų su ankstesnėmis. 

Galimas toks jungimo būdas: pasirenkame vieną iš duotų taškų kaip atskaitos tašką O ir nubrėžiame iš jo spindulius į visus kitus taškus.

Paimame kažkokį spindulį, pavyzdžiui Oy, kaip pradinį ir surikiuojame spindulius pagal kampą (prieš laikrodžio rodyklę) sudaromą su pradiniu spinduliu. Kelis sutampančius spindulius laikykime vienu. 

Tada daugiakampį jungiame taip: brėžiame atkarpą nuo O iki tolimiausio taško pirmame spindulyje, toliau brėžiame atkarpą iki artimiausio taško nuo O antrame spindulyje, toliau vėl brėžiame atkarpą iki tolimiausio taško antrame spindulyje ir taip pereiname visus spindulius iki priešpaskutinio, o paskutiniuoju spinduliu pereiname nuo tolimiausio taško iki artimiausio ir sugrįžtame į tašką O.

Tokiu būdu visada gausime daugiakampį, nes atkarpos tarp gretimų spindulių ir spinduliuose nesikerta.

Rikiuojant taškus, gali kilti keblumų su geometrija. Tarkime, kad turime tris taškus: A, B ir O. Susitarkime, kad taškas O yra koordinačių pradžios taškas. Panagrinėkime, kaip nustatyti, kuris iš spindulių OA ar OB sudaro didesnį kampą su spinduliu Oy. O jei kampas yra vienodas – kuris iš taškų yra arčiau koordinačių pradžios taško. Bus patogu, jei tarp dviejų taškų įvesime sąryšį mažiau. Laikysime, kad A mažiau už B, jei kampas OyA < OyB arba, jei OyA ( OyB ir |OA| < |OB|. 

Pirmiausia patikriname abiejų taškų x koordinačių ženklus. Galimi trys atvejai:

– ženklai yra priešingi. Tai reiškia, kad abu taškai yra skirtingose spindulio Oy pusėse. Šiuo atveju A mažiau už B, jei A.x < 0.

– bent vieno taško koordinatė x lygi 0: A.x ( 0 arba B.x ( 0. Tai reiškia, kad bent vienas iš taškų yra ant y ašies. Šiuo atveju nesunku nustatyti, kada A mažiau už B.

– ženklai sutampa. T. y. abu taškai yra vienoje spindulio Oy pusėje. Matome, kad tg((’) ( A.x/A.y, o tg((’) ( B.x/B.y. Matome, kad juo mažesnis yra kampas OyA (26.6 pav. kampas (), tuo mažesnis yra to kampo tangentas. 

Pastaba. Ši savybė galioja tik tada, kai abu taškai yra toje pačioje spindulio pusėje.

function mažiau (O, A, B: taškas): boolean;

{ palyginamos taškų A ir B padėtys taško O atžvilgiu: }
{ Oy yra vertikalus spindulys į viršų nuo taško O, k(M) – kampas MOy matuojamas }

{ prieš laikrodžio rodyklę nuo Oy, čia M – taškas }
{ Taigi, P(A) < P(B), kai k(A) < k(B) arba kai k(A) = k(B) ir |OA| < |OB| }
  function ženklas (x: integer): integer;

  begin
    if x > 0 then ženklas := 1

             else if x < 0 then ženklas := -1

                           else ženklas := 0

  end; { ženklas }
begin
  A.x := A.x - O.x; A.y := A.y - O.y; { pereinama į koordinačių } 

  B.x := B.x - O.x; B.y := B.y - O.y; { sistemą su centru taške O }
  case ženklas (A.x) * ženklas (B.x) of
    -1: { A ir B priešingose vertikaliosios ašies pusėse }
        mažiau := (A.x < 0);

     0: { A arba B yra ant y ašies }
        mažiau := (B.x > 0) or (B.y < 0) and (A.x < 0) or
                  (B.x < 0) and (A.y < 0) or
               A.x = 0) and (B.x = 0) { abu taškai ant ašių }
               and (((A.y<B.y) or (B.y<0)) and (A.y>0) or
               (A.y > B.y) and (A.y < 0));

      1: { A ir B toje pačioje pusėje }
         mažiau := (A.y*B.x<A.x*B.y) or
                (A.y*B.x=A.x*B.y) and (abs(A.x)<abs(B.x))

   end
end;  { mažiau }
procedure sutvarkyti (var d: tdaugiakampis; n: integer);

{ daugiakampio taškai surikiuojami pagal sąryšį mažiau }

{ naudojant Greitojo rikiavimo metodą}
   procedure rūšiuok (ka, de: integer);

     var i, j: integer;

         x, y: taškas;

   begin
      i := ka; j := de; x := d[(ka + de) div 2];

      repeat
        while mažiau (d[0], d[i], x) do
          i := i + 1;

        while mažiau (d[0], x, d[j]) do
          j := j - 1;

        if i <= j then
          begin
            y := d[i];

            d[i] := d[j];

            d[j] := y;

            i := i + 1;

            j := j - 1;

          end
      until i > j;

      if ka < j then rūšiuok (ka, j);

      if i < de then rūšiuok (i, de)

   end; { rūšiuok }
begin
  rūšiuok (1, n)

end; { sutvarkyti }
procedure rikiuoti (var d: tdaugiakampis; 

                    n: integer; { duotieji taškai }
                    var galima: boolean 

                    { ar galima sudaryti daugiakampį });

{ jei daugiakampį sudaryti galima, taškai surikiuojami pagal kampą su }
{ pradiniu spinduliu }
begin
  if not ne_tiesėje (d,n) or (n<=2) { jei visi taškai vienoje tiesėje }
  then galima := false              { ar jų per mažai }
  else begin
         galima := true;

         sutvarkyti (d, n) { surikiuojami taškai }
       end;

end; { rikiuoti }
Surikiavus taškus nėra akivaizdu, su kuriuo spinduliu pirmiausia reikia sujungti tašką O 

Taigi, prieš spausdinant būtina patikrinti, ar nėra dviejų spindulių kampas tarp kurių viršija 180(. Jei yra, iš taško O būtina eiti į tašką, esantį viename šių spindulių.

function kitas (x, n: integer): integer;

begin
  if x = n

     then kitas := 1

     else kitas := x + 1

end; { kitas }
function atgal (x, n: integer): integer;

begin
  if x = 1

     then atgal := n

     else atgal := x - 1

end; { atgal }
function mažiau_už_180 (O, A, B: taškas): boolean;

begin
  A.x := A.x - O.x; A.y := A.y - O.y;

  B.x := B.x - O.x; B.y := B.y - O.y;

  mažiau_už_180 := (B.y*A.x > B.x*A.y) or
                 (B.y*A.x=B.x*A.y) and (A.x*B.x>-A.y*B.y)

end;

procedure rasti_pradžią (d: tdaugiakampis; n: integer;

                         var t: integer);

{ randa tašką, kurį reikia jungti su tašku O }
begin
  t := 1;

  while mažiau_už_180(d[0],d[t],d[kitas(t,n)])and(t<=n)do
    t := t + 1;

  t := kitas (t, n)

end;

procedure rezultatai (d: tdaugiakampis; n: integer;

                      galima: boolean);

{ išveda taškus tokia tvarka, kad jungiant juos paeiliui būtų }
{ gaunamas daugiakampis }
  var f: text;

  procedure rašyk_tašką (i: integer);

  begin
    writeln (f, d[i].x, ' ', d[i].y)

  end;

  var pirm, pask, i, a: integer;

begin { rezultatai }
  assign (f, 'DAUG.REZ');

  rewrite (f);

  if not galima

  then writeln (f, 'NEĮMANOMA')

  else 

   begin
     rasti_pradžią (d, n, a); { a – pirmasis pirmojo spindulio taškas }
     { ieškosime tolimiausio paskutinio spindulio taško }
     pask := atgal (a, n);

     { ieškosime pirmo to paties spindulio taško }
      pirm := pask;

      while ant_tiesės (d[0], d[pirm], d[pask]) do
        pirm := atgal (pirm, n);

      pirm := kitas (pirm, n);

      rašyk_tašką (0);

      { taškas O yra tarp taškų a ir pirm }
      while a <> pirm do
        begin
          rašyk_tašką (a);

          a := kitas (a, n)

        end;

      for i := pask downto pirm do
        rašyk_tašką (i);

   end;

  close (f)

end; { rezultatai }
Kadangi gan sudėtingas šio uždavinio išvedimas, pateikiame ir pagrindinę programos dalį. Trūksta tik duomenų įvedimo procedūros, kurią rasite kartu su testais. Duomenų įvedimas šiam uždaviniui – labai nesudėtingas.

program daugiakampis;

  const MAXN = 100; { didžiausias galimas taškų skaičius }
  type taškas = record
                  x, y: integer

                end;

       tdaugiakampis = array [0..MAXN] of taškas;

  var n: integer;

      d: tdaugiakampis;

      galima: boolean;

begin
  skaityti (d, n);  { perskaitomi pradiniai duomenys }
  rikiuoti (d, n, galima); { taškai surikiuojami pagal kampą }
                           { sudaromą su pradiniu spinduliu }
  rezultatai (d, n, galima)

end.

	Testo nr.
	Komentarai

	1
	Duotas tik vienas taškas

	2
	Duoti trys taškai, visi vienoje tiesėje

	3
	Galima sudaryti iškiląjį daugiakampį

	4
	Galima sudaryti paprastąjį neiškiląjį daugiakampį

	5
	Duoti visi taškai iš intervalo: 0 ( x ( 5, 1 ( y ( 6

